
1) Model for the sea surface temperature
model {
    trend[1]<-alpha+beta[1]*cos(2*pi*JDAYS[1]/365)+beta[2]*sin(2*pi*JDAYS[1]/365) # defining the first element of the series
    for(i in 2 : N) { # N is the number of days 
      T[i] ~ dnorm(mu[i], tau) # T is observed temperature
      mu[i] <- trend[i] + ar1 * (T[i-1] - trend[i-1]) # including autocorrelation (AR1)
	trend[i]<-alpha+beta[1]*cos(2*pi*JDAYS[i]/365)+beta[2]*sin(2*pi*JDAYS[i]/365) # seasonal trend
    }
    ar1 ~ dunif(-1.1,1.1) # Prior for autocorrelation parameters
    alpha~ dnorm( 0 , 1.0E-6 ) # Prior for alpha
    beta[1] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta1
    beta[2] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta2
    tau ~ dgamma( 0.001 , 0.001 ) # Prior for tau (tolerance for random variability around the expected temperature)
#derived
phi<-atan(beta[2]/beta[1]) # maximum
}
2) Model for the spawning probability
model {
#fish level
for (i in 1:N){ # number of fish sampled
	gsi[i]~dgamma(mu.i[i],r.i[i]) # observed GSI
	r.i[i]<-r[zz[i]] # parameter “rate” of the gamma of the individual (NOTE that it is conditional to the state of the individual, zz[i] (spawning or not)
	mu.i[i]<-mu[zz[i]] # parameter “shape” of the gamma of the individual (NOTE that it is conditional to the state of the individual, zz[i] (spawning or not)
	z[i]~dbern(pS[i]) # current state of the fish i
	zz[i]<-z[i]+1 # trick for setting zz (either 0 or 1) from the output of the Bernoulli distribution (either 1 or 2)
	logit(p.r[i])<-alpha+beta1*cos(2*pi*JDAYS[i]/365)+beta2*sin(2*pi*JDAYS[i]/365) # p.r is the probability of a female is at spawning state, given that it is sexually mature
	pS[i]<-p.mad[i]*p.r[i] # p.mad is the probability that a female is sexually mature (DATA; they are inferred from TL and morphotype, from empirical data)
}
#state level (spawning and non-spawning)
for (i in 1:2){ # 2 states; r and mu are the two parameters of a gamma distribution
	r[i] ~ dunif(0,100)
	mu[i] ~ dunif(0,100)
}
#Priors
    alpha~ dnorm( 0 , 1.0E-6 ) # Prior for alpha
    beta[1] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta1
    beta[2] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta2
#derived
phi<-atan(beta2/beta1)

3) Model for the feeding index
model {
#fish level
for (i in 1:N){ # N fishes 
	gut[i]~dgamma(shape[MUESTREO[i]],rate[MUESTREO[i]]) # shape and rate
} # MUESTREO is a given haul/sample
for (i in 1:M){ # M hauls/samples
	rate[i] ~ dunif(0,100) # Prior for rate
	shape[i] <-mu[i]*rate[i]
	mu[i]<-alpha+beta1*cos(2*pi*JDAYS[i]/365)+beta2*sin(2*pi*JDAYS[i]/365) # trend
}
#Priors
    alpha~ dnorm( 0 , 1.0E-6 ) # Prior for alpha
    beta[1] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta1
    beta[2] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta2
#derived
phi<-atan(beta2/beta1)
}
4) Model for the distance travelled (note that the model for the home range is identical)
model {
#fish level
for(i in 1:npeces){ # npeces is the number of fish
	trend[1,i]<-alpha0+alpha[i]+beta[1]*cos(2*pi*JDAYS[1]/365)+beta[2]*sin(2*pi*JDAYS[1]/365)+beta[3]*TL[i] # setting the value for the first element in the series
	for(j in 2:M){ # nested loop within each fish: M are time periods (10 days) 
		trend[j,i]<-alpha0+alpha[i]+beta[1]*cos(2*pi*JDAYS[j]/365)+beta[2]*sin(2*pi*JDAYS[j]/365)+beta[3]*TL[i] 
		mu[j,i]<-trend[j,i]+rho*(D[j-1,i]-trend[j-1,i]) # adding autocorrelation
		D[j,i]~dnorm(mu[j,i], tau)
      }
}
#priors for fish (intercept random effects)
for(i in 1:npeces){
     alpha[i]~dnorm(0,tau.alpha) # tau.alpha determines the between-fish variability in intercepts
}
#hyperpriors
alpha0~ dnorm( 0 , 1.0E-6 )  # Prior for the general intercept
for (i in 1:3){
	beta[i]~dnorm(0, 1.0E-6) # Priors for betas
}
tau~dgamma(0.001,0.001) # 
tau.alpha~dgamma(0.001,0.001) # Prior for tau.alpha (it determines the between-fish variability in intercepts)
rho~dunif(-1,1)  # Prior for the autocorrelation term
#derived
phi<-atan(beta[2]/beta[1])
}
5) Model for catchability
model {
for (i in 1:N){ # N is the number of gear deployments
	CAP[i] ~ dpois(mu.eff[i]) # CAP are captures (number of fish per fishing journey)
	abundance[i]~dgamma(shape[i],rate[i]) # PRIORS for expected abundance (hidden, unobserved variable). Shape and rate are the gamma parameters and they are inferred from Pita (2011).
	mu.eff[i]<-abundance[i]*p[i]  #  expected catches, given abundance and p
	logit(p[i])<-log(EFF[i])+alpha+beta1*cos(2*pi*JDAYS[i]/365)+beta2*sin(2*pi*JDAYS[i]/365)	
} # seasonal trend for p
#Priors
    alpha~ dnorm( 0 , 1.0E-6 ) # Prior for alpha
    beta[1] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta1
    beta[2] ~ dnorm( 0 , 1.0E-6 ) # Prior for beta2
#derived
phi<-atan(beta2/beta1)

